
*        CMOS SWITCHES
*                 ___       ___
*             ___|IN2|/|___|  _|_    CMOS AMP
*             |  |___|\| :   /1/F\
*   ___       |   ___    :   \___/ |\
*  |IN1|_____/|\_|IN1|   :     |___|+\     ____
*  |___|  |   |  |___|   :         |  \___|OUT1|
*         |   |          :         |  /   |____|
*         |   |   ___    :       __|-/
*         |__/|\_|IN1|/|___      | |/
*             |  |___|\|   |_____|
*    ___      |   ___
*   |IN2|_____|__|IN2|
*   |___|        |___|
*
*
*              ______        ______           ______  OUT2
*      VIN1   |      |OUT1  |  _   |VSUM     |      |                     ___
*      _____|\| \ /  |____|\|   /  |_______|\| \ /  |___/\  /\  /\_______|LP |
*    _|_    |/|  X   |    |/|  _\  |       |/|  X   |     \/  \/  _|_    |___|
*   /VIN\     | / \  |      |______|         | / \  |      ROUT   ___
*   \___/     |______|          ^ BSUM       |______|             _|_ COUT
*    _|_         ^  BMUX1      /_\              ^  BMUX2          ///
*    ///        /_\            _|__ VSIN       /_\
*                |_____       /_   \            |
*               _|_    |     // \   \           |
*         VCLK /_  \   |     \   \_//           |
*             /| |  \  |      \____/            |
*             \  |_|/  |       _|_              |
*              \___/   |       ///              |
*                |     |________________________|
*               _|_
*               ///
*
*

VSIN        VSIN  0    SIN( 0 1.3u 100)
VCLK        VCLK  0    PULSE( -.7 .7 1n 1n 1n .1m .2m)
VIN1        VIN1  0    DC    1.1u
BMUX1       OUT1  0    V  = u(V(VCLK))*V(VIN1) -1*u(-1*V(VCLK))*V(VIN1)
BSUM        VSUM  0    V  = V(OUT1) + V(VSIN)
BMUX2       OUT2  0    V  = u(V(VCLK))*V(VSUM) -1*u(-1*V(VCLK))*V(VSUM)
ROUT        LP    OUT2 3k
COUT        LP    0    3u

.control
set pensize = 2



tran        1u   10m    0  1u
plot        v(vin1) v(vsin) 
plot        v(vsum)
plot        v(out2) 
plot        v(vin1)  v(lp) v(vsin)

.endc

.end

The Chopper Amplifier takes advantage of the availability 
of effectively perfect switches. While vacuum tubes and 
transistors introduces errors like DC offset and 1/f noise,
switches come close to being just either on or off. 

*        CMOS SWITCHES
*                 ___       ___
*             ___|IN2|/|___|  _|_    CMOS AMP
*             |  |___|\| :   /1/F\
*   ___       |   ___    :   \___/ |\
*  |IN1|_____/|\_|IN1|   :     |___|+\     ____
*  |___|  |   |  |___|   :         |  \___|OUT1|
*         |   |          :         |  /   |____|
*         |   |   ___    :       __|-/
*         |__/|\_|IN1|/|___      | |/
*             |  |___|\|   |_____|
*    ___      |   ___
*   |IN2|_____|__|IN2|
*   |___|        |___|

Chopping the input signal using switches is really just 
modulating or multiplexing a low frequency signal up to 
the chopping frequency. If the chopping frequency is high
enough, the modulated input signal is well above the offset
and 1/f noise of any amplifier which follows the chopping
switches. This is the trick where signal gets placed where
it can be distinguished from everything else. 



*
*              ______        ______           ______  OUT2
*      VIN1   |      |OUT1  |  _   |VSUM     |      |                     ___
*      _____|\| \ /  |____|\|   /  |_______|\| \ /  |___/\  /\  /\_______|LP |
*    _|_    |/|  X   |    |/|  _\  |       |/|  X   |     \/  \/  _|_    |___|
*   /VIN\     | / \  |      |______|         | / \  |      ROUT   ___
*   \___/     |______|          ^ BSUM       |______|             _|_ COUT
*    _|_         ^  BMUX1      /_\              ^  BMUX2          ///
*    ///        /_\            _|__ VSIN       /_\
*                |_____       /_   \            |
*               _|_    |     // \   \           |
*         VCLK /_  \   |     \   \_//           |
*             /| |  \  |      \____/            |
*             \  |_|/  |       _|_              |
*              \___/   |       ///              |
*                |     |________________________|
*               _|_
*               ///  
The simulation shows how this is done by applying a 1.1uV
DC voltage in front of the input choppers and summing the
output  with a 100Hz sign wave to represent 1/f noise. 

The sum of the modulated DC signal and the AC signal
are shown below. 



When this signal get remodulated by the same clock,
the DC signal will return back to its original 
form. But the AC signal is now being modulated by
the same chopping clock frequency.  

*
*              ______        ______           ______  OUT2
*      VIN1   |      |OUT1  |  _   |VSUM     |      |                     ___
*      _____|\| \ /  |____|\|   /  |_______|\| \ /  |___/\  /\  /\_______|LP |
*    _|_    |/|  X   |    |/|  _\  |       |/|  X   |     \/  \/  _|_    |___|
*   /VIN\     | / \  |      |______|         | / \  |      ROUT   ___
*   \___/     |______|          ^ BSUM       |______|             _|_ COUT
*    _|_         ^  BMUX1      /_\              ^  BMUX2          ///
*    ///        /_\            _|__ VSIN       /_\
*                |_____       /_   \            |
*               _|_    |     // \   \           |
*         VCLK /_  \   |     \   \_//           |
*             /| |  \  |      \____/            |
*             \  |_|/  |       _|_              |
*              \___/   |       ///              |
*                |     |________________________|
*               _|_
*               ///  
If signal is gained up high enough before applying to the 
second modulator, a standard four quad multiplier circuit
will do the job. 



The desired 1.1uV VIN1 signal can be extracted by applying a
low pass filter. 

The chopper effectively passes the desired input signal
through unchanged while it modulate all other undesirable
signals up to the chopping frequency where they can be
lowpass filtered out. 


